

MARVEL SUPER HEROES™

DF  
DONALD I. FINE, INC.

# COMPUTER FUN

BOOK TWO



ALL PROGRAMS  
RUN ON


7

TOP MICROS

\$7.95

20 ACTION-FILLED PROGRAMS





**ENTER A  
WORLD OF  
SUPER COMPUTER  
ADVENTURE!**


Join the MARVEL SUPER HEROES™ as they valiantly battle the forces of evil and destruction. Use your trusty computer to run 21 super programs that make you a part of the action. Plug in your computer and get set to load it with codes, computer puzzles, games, graphics, and loads of fun!!!



**MARVEL SUPER HEROES™**

# COMPUTER FUN

## BOOK TWO



By Richard Guaraldo and Susan M. Zakar  
Continuity by Dwight Jon Zimmerman



A Parachute Press Book  
Donald I. Fine, Inc.



## Acknowledgments

---

The authors wish to thank the faculty and students of the Four Seasons Elementary School in Gambrills, Maryland, for their support in the preparation and testing of these programs.

Thanks to my wife, Clarena, who has been willing to be a "computer widow" as the material for this book was assembled. She has spent many hours providing inspiration and reviewing my creations. Also to our daughter, Patricia, who has been my quality assurance department (guinea pig). She has personally placed her stamp of approval on my work—RG

Thanks to my husband, Joe, who is a real "computer wizard" and whose love, patience, and gentle instruction helped me to understand the spellbinding magic of computers. He spent many a lonely night while I entered incantations (aka programs) into my microcomputers. Without his support this book would not have been possible. Thanks also to my young son, David, who almost always went to bed on time and slept well, which so greatly increased the time I had to program—SMZ

Parachute Press wishes to thank the creative staff at Marvel Comics for all their help and advice. And special thanks to Michael Z. Hobson, Vice President, Publishing, Marvel Comics Group without whom this project would not have been possible.

---

Illustrations copyright © 1984 MARVEL COMICS GROUP, a division of Cadence Industries Corporation.  
Text copyright © 1984 Parachute Press, Inc.

Spider-Man, Iron Man, Captain America, Thor, Mr. Fantastic, The Thing, The Invisible Girl, The Human Torch, Doctor Doom, Professor X, The X-Men, Cyclops Wolverine, Storm, Colossus, Nightcrawler, Ariel, Doctor Strange, Daredevil, Stilt Man, Mole Man, Modok, Valkyrie, Hellcat, Gray Gargoyle, and the distinctive likenesses thereof are Trademarks of the Marvel Comics Group, a division of Cadence Industries Corporation and are used with permission.

Marvel Super Heroes™ is a trademark of the Marvel Comics Group, a division of Cadence Industries Corporation. Used with permission.

All rights reserved, including the right of reproduction in whole or in part in any form. Published in the United States of America by Donald I. Fine, Inc. and in Canada by Fitzhenry & Whiteside, Ltd.

ISBN: 0-917657-06-3

Manufactured in the United States of America

10 9 8 7 6 5 4 3 2 1

VIC 20 is a trademark of Commodore Computer.



PARACHUTE PRESS  
DEDICATES THIS BOOK  
TO RUSS D'ANNA  
AND MARY DEE ENGLISH






# ATTENTION, ADVENTURE LOVERS!!

**YOUR COMPUTER IS ABOUT TO TURN YOU INTO A SUPER HERO.' WITH THE BASIC PROGRAMS IN THE BOOK, YOU WILL FEEL AS IF YOU ARE SCALING WALLS WITH THE AMAZING SPIDER-MAN.' YOU WILL HAVE THE SUPER SENSES OF DAREDEVIL. THE INCREDIBLE MIND OF REED RICHARDS, LEADER OF THE FANTASTIC FOUR. THE MAGICAL POWERS OF THOR AND DOCTOR STRANGE. YOU WILL OUTWIT THE VILLAINOUS HOBGOBLIN, TANGLE WITH DOCTOR DOOM AND GO UP AGAINST THE RUTHLESS KINGPIN.'**

**ALL YOU HAVE TO DO IS TYPE THESE PROGRAMS INTO YOUR COMPUTER. THEN ENTER THE MAGIC WORD *RUN*, AND YOU'LL BE OFF INTO A WORLD OF DANGER, EXCITEMENT, HEROISM AND VILLAINY.**

**BE CAREFUL. YOU MUST FOLLOW ALL THE INSTRUCTIONS VERY CAREFULLY--OR YOUR CAREER AS A SUPER HERO WILL BE ALL TOO BRIEF. START WITH THE *HOW TO USE THIS BOOK* SECTION ON THE NEXT PAGE. IT HAS THE IMPORTANT INFORMATION YOU REALLY NEED. WHAT ARE YOU WAITING FOR? *THE MARVEL SUPER HEROES™ CHARACTERS AND A WORLD OF COMPUTER FUN ARE WAITING FOR YOU!***



All the programs in this book have been designed to run in the BASIC programming language on the **IBM PC, PCjr, Commodore 64, VIC 20, Radio Shack Model III, Radio Shack Color Computer (with extended BASIC), Apple II, Apple II+, and Apple IIe, IIC.**

Getting the same BASIC programs to run on many different microcomputers is a difficult task, especially if you want to do anything more than compute and print. Each microcomputer may have a different way of doing things, like clearing the screen, positioning the cursor, or even generating a random number. We wanted to make it as simple as possible. That's why we created the "900-Lines."

Now don't panic! There are not NINE HUNDRED lines. There are less than ten. They have line numbers from 900 to 990 (the Radio Shack III and Color Computer have a line 50, too). Each microcomputer we have listed has its own set of 900-Lines. We've explained them in detail below. The 900-Lines are the only lines that change between computers. All the rest of the BASIC program lines are exactly the same for an IBM PC, an APPLE, COMMODORE 64, VIC 20, Radio Shack III, or Color Computer and PCjr. Just use the ten or so lines for your computer with each program, and PRESTO, it runs! Simple enough? Let's make it even simpler . . .

Most of the 900-Lines don't even change from program to program. That means that you can type them in ONCE, save them just like a BASIC program, and then use them for program after program after program! Here's how to do it:

First, make sure your computer is in BASIC. Then type in the 900-Lines for your microcomputer from the listings. Next, save these lines the same way you would save any BASIC program to tape or disk. We especially like the name 'N900' for this program. Now, BEFORE you type in any of the programs in the book, LOAD the 900-Lines. Type in the rest of the program lines, along with any special changes noted (always lines 970 to 990) and your program is complete. Save the whole new program under its name if you want to load and run it again later.



Here's what all the 900-Lines do:

<b>900</b>	<b>CLEAR SCREEN</b>	This subroutine clears the screen. It does not change from program to program.
<b>910</b>	<b>POSITION CURSOR</b>	This subroutine moves the cursor to a specific place on the screen. The variable VT (Vertical Tab) sets the position up or down on the screen. The variable HT (Horizontal Tab) sets the position across the screen. This line does not change from program to program. VT and HT must be properly set before calling this subroutine.
<b>920</b>	<b>WAIT LOOP</b>	This subroutine is called a WAIT LOOP because all it does is go around in a loop the number of times specified by WT. This subroutine is useful for slowing down programs that run too fast in places. It is the same for all programs and all machines.
<b>930</b>	<b>RANDOM NUMBER GENERATOR</b>	This subroutine returns a RANDOM number from 1 to the value of the variable RX. The number is returned in the variable RD. This line does not change from program to program.
<b>940-950</b>	<b>KEYPRESS</b>	This is the KEYPRESS routine. It detects whether a key has been pressed on the keyboard. If one has been pressed, the character is returned in the string variable KY\$. If no key has been pressed, KY\$ is set to NU\$, or CHR\$ (0). These lines do not change from program to program.
<b>960</b>	<b>SETUP</b>	This is the SETUP line. It sets the variables for Screen Height (SH) and Screen Width (SW) and any other variables required by the other 900-Lines. It also makes a call to line 970 where program-specific parameters are set up. This line does not change from program to program.
<b>970-990</b>	<b>PROGRAM SPECIFIC SETUP</b>	These program lines are for setting up variables and parameters which are peculiar for any one program. These lines <i>MAY CHANGE</i> from program to program. They may seed a random number generator, or set the speed for the wait loop or things like that. All three lines are not always used, but they are available just in case. The last line actually used is the one that ends with a RETURN statement.



**DON'T FORGET  
YOUR 900-LINES!  
THEY CUSTOMIZE  
THE PROGRAMS  
FOR YOUR  
COMPUTER.**



After you have loaded the 900-Lines, all you have to do is input the BASIC program in the text. Be very careful to type the listing just as it appears. Even an extra space or a comma can prevent the program from running correctly. Also remember to use ALL CAPITAL LETTERS when you are inputting data or hitting a key to play a game.

Some of the programs have been designed to give you the challenge of debugging them. That means that you must change something in the program so that it will run correctly. The instructions will tell you when you've come across one of these programs. And if you get stumped as to how to fix it, there's help in the Super Hint Section in the back of the book.

Have fun!



### Use the following lines for the COMMODORE 64

```
900 PRINT CHR$(147):RETURN
910 POKE XT,HT-1:POKE YT,VT-1:POKE FG,0:SYS PL:RETURN
920 FOR WS=1 TO WT:NEXT:RETURN
930 RD=INT(RX*RND(1)+1):RETURN
940 GET KY$:IF KY$="" THEN KY$=NU$
950 RETURN
960 XT=782:YT=781:FG=783:PL=65520:SW=40:SH=24:NU$=CHR$(0):
    GOSUB 970:RETURN
970 RD=RND(-TI):RETURN
```

### Use the following lines for the COMMODORE VIC 20

```
900 PRINT CHR$(147):RETURN
910 POKE XT,HT-1:POKE YT,VT-1:POKE FG,0:SYS PL:RETURN
920 FOR WS=1 TO WT:NEXT:RETURN
930 RD=INT(RX*RND(1)+1):RETURN
940 GET KY$:IF KY$="" THEN KY$=NU$
950 RETURN
960 XT=782:YT=781:FG=783:PL=65520:SW=22:SH=22:NU$=CHR$(0):GOSUB
    970:RETURN
970 RD=RND(-TI):RETURN
```

### Use the following lines for the RADIO SHACK COLOR COMPUTER

```
50 CLEAR 2000
900 CLS:RETURN
910 HZ=INT(HT-1+(VT-1)*SW+0.5):PRINT@ HZ,"";:RETURN
920 FOR WS=1 TO WT:NEXT:RETURN
930 RD=INT(RND(RX)):RETURN
940 KY$=INKEY$:IF KY$="" THEN KY$=NU$
950 RETURN
960 NU$=CHR$(0):SW=32:SH=16:GOSUB 970:RETURN
970 RETURN
```



### Use the following lines for the RADIO SHACK MODEL III

```
50 CLEAR 2000
900 CLS:RETURN
910 HZ=INT(HT-1+(VT-1)*SW+0.5):PRINT@ HZ,"";:RETURN
920 FOR WS=1 TO WT:NEXT:RETURN
930 RD=INT(RND(RX)):RETURN
940 KY$=INKEY$:IF KY$="" THEN KY$=NU$
950 RETURN
960 NU$=CHR$(0):SW=64:SH=16:GOSUB 970:RETURN
970 RETURN
```

### Use the following lines for the IBM PC and IBM PCjr

```
900 CLS:RETURN
910 LOCATE VT,HT:RETURN
920 FOR WS=1 TO WT:NEXT:RETURN
930 RD=INT(RX*RND(1)+1):RETURN
940 KY$=INKEY$:IF KY$="" THEN KY$=NU$
950 RETURN
960 SW=40:SH=24:NU$=CHR$(0):GOSUB 970:RETURN
970 RANDOMIZE(VAL(RIGHT$(TIME$,2))):RETURN
```

### Use the following lines for the APPLE II, APPLE II+, APPLE //e, and APPLE //c.

```
900 HOME:RETURN
910 VTAB VT:HTAB HT:RETURN
920 FOR WS=1 TO WT:RETURN
930 RD=INT(RX*RND(1)+1):RETURN
940 KY$=NU$:KY=PEEK(KZ):IF KY<128 THEN RETURN
950 KY$=CHR$(KY-128):POKE KW,0:RETURN
960 NU$=CHR$(0):SW=40:SH=24:KZ=-16384:KW=-16368:GOSUB 970:RETURN
970 RETURN
```



# IRON MAN



## TO THE RESCUE!

IRON MAN MUST STOP THOSE  
ENEMY MISSILES FROM HIT-  
TING THE SCHOOL! TURN THE  
PAGE TO FIND OUT HOW YOU  
CAN BECOME IRON MAN  
AND DEFEND THE SCHOOL  
AGAINST THE DEADLY MIS-  
SILES.



# Program 1

## Bombs Away

Input the following program in your computer and run it.

Hit U for Up, D for Down, L for Left, R for Right.

Fire the repulsor rays to defend the school house from the blob bullets.

The computer will compute your defense average. An average of 65% will help Iron Man succeed.

Load your 900 lines before typing this program.

```
100 REM BOMBS AWAY
110 GOSUB 960:GOSUB 900
120 DIM HX(12),VY(12),L$(12)
130 F$(1)="U":F$(2)="D":F$(3)="L":F$(4)="R"
140 FOR I=1 TO 12
150 READ HX(I),VY(I),L$(I)
160 VT=VY(I):HT=HX(I):GOSUB 910
170 L$(1)=CHR$(94)
180 PRINT L$(I);
190 NEXT
200 RX=4:GOSUB 930:N=RD+8
210 I=N:VT=VY(I):HT=HX(I):GOSUB 910
220 PRINT "O";
230 GOSUB 370
240 VT=VY(I):HT=HX(I):GOSUB 910
250 IF HT=0 THEN 270
260 PRINT"*";
270 VT=SH-1:HT=1:GOSUB 910:PRINT" ";
280 WT=S:GOSUB 920
290 VT=VY(I):HT=HX(I):GOSUB 910
300 PRINT " ";
310 A=A+1
320 IF A<25 THEN 200
330 GOSUB 900
340 PRINT "YOU FOUGHT OFF":PRINT SC;" BOMBS."
```



```

350 PRINT "YOUR DEFENSE AVERAGE":PRINT "IS ";INT(SC/25*100);"%
360 END
370 T=S:HF=0
380 GOSUB 940
390 IF KY$=NU$ THEN 420
400 IF KY$=F$(N-8) THEN SC=SC+1:GOSUB 450:HF=1
410 RETURN
420 T=T-1
430 IF T>0 THEN 380
440 RETURN
450 VT=SH-1:HT=1:GOSUB 910:PRINT "*HIT*"
460 WT=INT(S/2):GOSUB 920:RETURN
500 DATA 7,5,!,7,6,!,7,8,!,7,9,V
510 DATA 5,7,<,6,7,-,8,7,-,9,7,>
520 DATA 7,2, ,7,12, ,2,7, ,12,7,
530 DATA 7,5,!,7,6,!,7,8,!,7,9,V

```

---

# BOMBS AWAY CHANGES FOR SPECIFIC MACHINES

## 1. TRS 80 MODEL III

```
970 S=50: RETURN
```

## 2. COMMODORE 64 AND VIC-20

```
970 RD=RND(-TI):S=100: RETURN
```


## 3. IBM PC AND PCjr

```
970 RANDOMIZE VAL(RIGHT$(TIME$,2)):S=100:RETURN
```

## 4. APPLE AND COLOR COMPUTER

```
970 S=100:RETURN
```





## ***CAPTAIN AMERICA'S*** **NAME FINDER**

SOMEONE IS ABOUT  
TO ATTACK  
CAPTAIN AMERICA!

**WHO CAN IT BE?**

YOU CAN USE THE  
PROGRAM ON THE  
NEXT PAGE TO HELP  
CAP. FIGURE OUT THE  
NAME OF HIS ENEMY.



# Program 2

## Guess Who

Input the following program in your computer and run it.

Try and guess the name of Captain America's enemy.

This program works like "Hangman."

Load your 900 lines before typing this program.


```
100 REM GUESS WHO
110 GOSUB 900 :GOSUB 960
120 RX=1:GOSUB 930:RN=RD
130 FOR I=1 TO RN:READ S$:NEXT
140 SX$="":FOR I=1 TO LEN(S$)
150 SX$=SX$+CHR$(ASC(MID$(S$,I,1))-1)
160 NEXT I:S$=SX$
170 FOR I=1 TO LEN(S$)
180 CH$=MID$(S$,I,1)
190 IF CH$=>"A" AND CH$<="Z" THEN CH$="-":CNT=CNT-1
200 SS$=SS$+CH$:NEXT
210 GOSUB 900
220 PRINT "GUESSES: ";G:PRINT " ERRORS: ";ER
230 HT=1:VT=5:GOSUB 910
240 PRINT SS$
250 IF CNT=0 THEN 480
260 VT=10:HT=1:GOSUB 910
270 PRINT "LETTER? >";
280 GOSUB 940 :IF KY$=NU$ THEN 280
290 G=G+1:R=0:PRINT KY$
300 FOR I=1 TO LEN(S$)
310 IF MID$(S$,I,1)<>KY$ THEN 440
320 S1$="":IF I=1 THEN 340
330 S1$=LEFT$(S$,I-1)
340 S2$="":IF I=LEN(S$) THEN 360
350 S2$=RIGHT$(S$,LEN(S$)-I)
360 S$=S1$+"."+S2$
370 CNT=CNT+1
380 S1$="":IF I=1 THEN 400
```



```
390 S1$=LEFT$(SS$,I-1)
400 S2$="":IF I=LEN(SS$) THEN 420
410 S2$=RIGHT$(SS$,LEN(SS$)-I)
420 SS$=S1$+KY$+S2$
430 R=1
440 NEXT I
450 IF R=0 THEN ER=ER+1
460 IF ER>5 THEN PRINT:PRINT"TOO MANY ERRORS!":STOP
470 GOTO 210
480 VT=SH-4:HT=1:GOSUB 910
490 PRINT"FINISHED"
500 END
510 DATA "HSBZ!HBSHPZMF"
```



# MENACE OF THE GRAY GARGOYLE



THE GRAY GARGOYLE CAN TURN ANYTHING THAT HE TOUCHES TO STONE. IT LOOKS AS IF HE'S GOT CAP TRAPPED. BUT MAYBE NOT! CAPTAIN AMERICA MAY BE ABLE TO DART THROUGH THE GAPS IN THE FENCE BEFORE THEY ARE TURNED TO SOLID ROCK. **WILL HE BE QUICK ENOUGH?** THE PROGRAM ON THE NEXT PAGE WILL TEST YOUR SPEED AND QUICKNESS. SEE HOW MANY TIMES YOU CAN CRASH THROUGH THE GARGOYLE'S FENCE.





# Program 3

## Crasher

Input the following program in your computer and run it. See if you can help Captain America dart through the fence before it turns to stone.

The exclamation point on the screen moves up by itself. Hit J to move left and K to move right.

Keep your eye on the fence. Good luck!

Load your 900 lines before typing this program.

```
100 REM CRASHER
110 GOSUB 900
120 GOSUB 960:K$="K":KY$=K$
130 FOR I=1 TO SW
140 L$=L$+"-"
150 NEXT I
160 VL=INT(SH/4)
170 VP=SH
180 HP=INT(SW/2)
190 RX=SW-5:GOSUB 930:R1=RD+1
200 RX=2:GOSUB 930:R2=RD:IF R2=2 THEN R2=-1
210 R1=R1+R2*ABS((R1+R2>1)*(R1+R2<SW-5))
220 M$=LEFT$(L$,R1-1)+" "+MID$(L$,R1+4)
230 FOR I=1 TO 5
240 VT=VL:HT=1:GOSUB 910
250 PRINT M$
260 K1$=KY$:GOSUB 940:IF KY$=NU$ THEN KY$=K1$
270 IF ASC(KY$)>=ASC("K") THEN HP=HP+ABS(HP<SW-2):GOTO 290
280 HP=HP+ABS(HP>2)*-1
290 VT=VP:HT=HP:GOSUB 910
300 PRINT " ! ";
310 NEXT
320 VT=VP:HT=HP-1:GOSUB 910
330 PRINT " ";
340 VP=VP-1
350 IF (VP=VL)*((R1<HP)+(R1>HP+3)) THEN 380
```



```
360 IF VP=VL THEN PRINT :PRINT"YOU WIN!":END
370 GOTO 200
380 PRINT "***CRASH***"
390 WT=WU:GOSUB 920
400 GOSUB 900:GOTO 160
```

---

CRASHER  
CHANGES FOR SPECIFIC COMPUTERS

1. TRS 80 MODEL III

```
970 WU=250:RETURN
```

2. COMMODORE 64 and VIC 20

```
970 WU=500:RD=RND(-TI):RETURN
```

3. IMB PC and PCjr

```
970 WU=500:RANDOMIZE VAL(RIGHT$(TIMES$,2)):RETURN
```

4. COLOR COMPUTER AND APPLES


```
970 WU=500:RETURN
```



# KINGPIN'S DECODER

THE SECRET  
NUMBER IS  
14!

SPIDER-MAN HAS SNEAKED INTO THE KINGPIN'S SECRET COMPUTER ROOM. HE HAS FOUND A CODED MESSAGE. BUT HE CANNOT READ IT. YOU MUST DECODE IT FOR HIM. THE PROGRAM ON THE NEXT PAGE WILL HELP YOU DO IT.



1. .MQFSA MA HO GW BOA-FSRWDG

2. - HVUWBCH HVUWBRWA MP

3. BWDUBWY .FSJWF HGOS SVH BW GSVGWT SVH UBWSST SP ZZWK SV

## Program 4 Kingpin's Decoder

Input the following program in your computer and run it. Enter the secret number from the preceding page. Then type in the coded message one line at a time. Type STOP when you've finished.

Load your 900 lines before typing this program.

```
100 REM KINGPIN'S DECODER
110 GOSUB 960:GOSUB 900
120 IF M$="STOP" THEN 270
130 INPUT "SECRET NUMBER (1..26) ";N
140 INPUT "MSG->";M$
150 IF M$="STOP" THEN 270
160 FOR I=LEN(M$) TO 1 STEP -1
170 C$=MID$(M$,I,1)
180 IF ((C$<"A")+ (C$>"Z")) THEN 220
190 V=ASC(C$)-N
200 IF V<ASC("A") THEN C$=CHR$(ASC("Z")-ASC("A")+V+1):GOTO 220
210 C$=CHR$(ASC(C$)-N)
220 PRINT C$;
230 NEXT I
240 PRINT
250 M$=""
260 GOTO 140
270 END
```



# SPIDEY'S MESSAGE



The fish  
didn't want me  
--- too stringy. Great  
computer. But you  
need more  
games.  
Spider-Man

SPIDEY WANTS TO LET THE KINGPIN KNOW THAT HE'S  
**STILL ALIVE AND WEB-SLINGING.** USE THE PROGRAM ON  
THE NEXT PAGE TO ENCODE SPIDEY'S MESSAGE SO  
SPIDEY CAN LEAVE IT AS A LITTLE SURPRISE FOR KINGPIN.



# Program 5

## Kingpin's Encoder

Input the following program in your computer and run it. Remember the secret number. Then enter the message on the preceding page one line at a time to encode Spidey's message.

You can use these programs to encode and decode your own secret messages.

Load your 900 lines before typing this program.

```
100 REM KINGPIN'S ENCODER
110 GOSUB 960:GOSUB 900
120 IF M$="STOP" THEN 270
130 INPUT "SECRET NUMBER (1..26) ";N
140 INPUT "MSG->";M$
150 IF M$="STOP" THEN 270
160 FOR I=LEN(M$) TO 1 STEP -1
170 C$=MID$(M$,I,1)
180 IF ((C$<"A")+(C$>"Z")) THEN 220
190 V=ASC(C$)+N
200 IF V>ASC("Z") THEN C$=CHR$(V-26-N)
210 C$=CHR$(ASC(C$)+N)
220 PRINT C$;
230 NEXT I
240 PRINT
250 M$=""
260 GOTO 140
270 END
```



# HAPPY BIRTHDAY, VALKYRIE!

HELLCAT AND HER FRIENDS IN  
THE DEFENDERS ARE GIVING  
VALKYRIE A **SURPRISE BIRTHDAY  
PARTY!** TURN THE PAGE FOR A  
PROGRAM THAT WILL MAKE YOUR  
OWN COMPUTER SCREEN INTO  
A MARQUEE!





# Program 6

## Marquee

Input the following program in your computer and run it.

Put any message you want to in Line 140 and watch it turn into a marquee. Just like Times Square!

Load your 900 lines before typing this program.

```
100 REM MARQUEE
110 GOSUB 960
120 GOSUB 900
130 READ NM$
140 DATA "HAPPY BIRTHDAY, VALKYRIE"
150 IF LEN(NM$)>SW-1 THEN 170
160 NM$=NM$+"....."
170 N$=LEFT$(NM$,SW-1)
180 GOSUB 900
190 WT=WU: GOSUB 920
200 FOR I=1 TO LEN(N$)
210 VT=5:HT=2:GOSUB 910
220 AA$="":IF LEN(N$)=I THEN 240
230 AA$=RIGHT$(N$,LEN(N$)-I)
240 PRINT AA$;LEFT$(N$,I)
250 VT=7:HT=2:GOSUB 910
260 AA$="":IF LEN(N$)=I THEN 280
270 AA$=LEFT$(N$,LEN(N$)-I)
280 PRINT RIGHT$(N$,I);AA$
290 WT=WU:GOSUB 920
300 NEXT
310 GOTO 200
```



MARQUEE  
CHANGES FOR SPECIFIC MACHINES

ALL MACHINES

970 WU=100:RETURN



# TRAINING WITH PROFESSOR X



PROFESSOR X HAS A NEW TEST TO MEASURE THE X-MEN'S RESPONSE SPEED. TURN THE PAGE TO SEE HOW YOU CAN PROGRAM YOUR COMPUTER TO TRY THE SAME TEST. **AND, WHATEVER YOU DO, DON'T FAIL! GOOD LUCK!** YOU CAN RECORD YOUR SCORES ON THE SUPERScore SHEET ON PAGE 36.



# Program 7

## Type Training

Input the following program in your computer and run it.

Start first with ten letters. See how fast you can type ten letters. As the letter comes up on the screen, hit that letter. The computer will tell you your score.

The number of units is the time it took you to type the letters.

Record your score on the superscore sheet. Now try again with ten letters, and see if you can type faster. Then try fifteen letters, twenty-five letters, as many letters as you can.

Use the superscore sheet to record your scores. With concentration, you should be able to type faster and faster.

**Load your 900 lines before typing this program.**

```
100 REM TYPE TRAINING
110 GOSUB 960
120 GOSUB 900
130 GOSUB 350
140 GOSUB 900
150 X=0:T=0:AL=ASC("A")-1
160 RX=26:GOSUB 930:C=RD+AL
170 VT=INT(SH/2): HT=INT(SW/2): GOSUB 910
180 PRINT CHR$(C)
190 GOSUB 940:K=ASC(KY$)
200 IF K=C THEN 230
210 T=T+1
220 GOTO 190
230 X=X+1
240 IF X>=Z THEN 260
250 GOTO 160
260 GOSUB 900
```



```
270 PRINT
280 PRINT "-----"
290 PRINT "SPEED: ";X;" LETTERS":PRINT"IN ";T;" UNITS"
300 PRINT "-----"
310 PRINT
320 INPUT "DO IT AGAIN?";Y$
330 IF Y$="Y" THEN 120
340 END
350 INPUT "HOW MANY LETTERS";Z
360 INPUT "HIT RETURN WHEN READY";C$
370 RETURN
```



# Superscore Sheet

**DATE & TIME**

**NUMBER OF LETTERS**

**UNITS ELAPSED**




# IRON MAN'S ROBOT RETRIEVER

IRON MAN MUST GET MODOK'S SECRET PLANS TO CONQUER THE WORLD! BUT MODOK IS SENDING A PSIONIC BEAM THAT IS SCRAMBLING THE ROBOT'S PROGRAMMING. **EVERY TIME THE ROBOT TRIES TO GET THE PLANS, IT WALKS STRAIGHT INTO A TRAP.** HELP HIM OUT! ENTER THE PROGRAM ON THE NEXT PAGE. THEN TRY TO OVERRIDE MODOK'S JAMMING AND MODIFY THE PROGRAM.





# Program 8

## Robot

Input the following program in your computer and run it.

You have to help the robot get the plans and stop it from falling into the trap.

List the program and see if you can figure out how Modok jammed the program to

trap the robot. Modify the program and run it again to watch the robot capture the plans. Hint: Try to figure out which part of the program controls the direction the robot walks.

If you need help in debugging the program, consult page 90 in the Hint Section for help.

Load your 900 lines before typing this program.

```
100 REM ROBOT
110 GOSUB 960:GOSUB 970
120 DIM A$(2,10)
130 FOR I=1 TO 2:FOR J=1 TO 8:READ A$(I,J):NEXT J,I
140 GOSUB 900
150 D=1: REM DIRECTION
160 GOSUB 350
170 VT=TV:HT=2:VV=VT
180 FOR I=VV TO 1 STEP -1
190 IF JJ=1 THEN JJ=2:GOTO 210
200 JJ=1
210 FOR J=1 TO 8:VT=I+J-1:GOSUB 910:PRINT A$(JJ,J);:NEXT
220 HT=HT+D
230 WT=100:GOSUB 920
240 NEXT
250 FOR I=1 TO VV
260 IF JJ=1 THEN JJ=2:GOTO 280
270 JJ=1
280 FOR J=1 TO 10:VT=I+J-1:GOSUB 910:PRINT A$(JJ,J);:NEXT
290 HT=HT+D
300 WT=100:GOSUB 920
310 NEXT
320 IF HT < 10 THEN 340
330 GOSUB 900:PRINT"GOTCHA!":END
340 GOSUB 900:PRINT"SAFE ":END
350 GOSUB 900:PRINT TAB(XV);"PLANS "
360 VT=11:HT=11:GOSUB 910:PRINT
370 PRINT "E";TAB( XX);"T"
```



```

380 PRINT "X";TAB( XX);"R"
390 PRINT "I";TAB( XX);"A"
400 PRINT "T";TAB( XX);"P":RETURN
410 STOP
420 DATA "      "
430 DATA "    O * "
440 DATA "  ***  "
450 DATA " * *   "
460 DATA "   **  "
470 DATA " * *   "
480 DATA " *    "
490 DATA "      "
500 DATA "      "
510 DATA " * O    "
520 DATA "  ***  "
530 DATA "   * *   "
540 DATA "   **   "
550 DATA " * *   "
560 DATA "   *    "
570 DATA "      "

```

---

## ROBOT CHANGES FOR SPECIFIC COMPUTERS

### 1. COMMODORE 64, APPLE, IBM PC and IBM PCjr

```
970 TV=12:XX=30:XV=14: RETURN
```

### 2. VIC-20

```
970 TV=7:XX=18:XV=8:RETURN
```

### 3. TRS 80 MODEL III and COLOR COMPUTER

```
970 TV=6:XX=19:XV=6:RETURN
```







# **THE STAR POWER OF THOR!**

**THOR CAN CREATE HUGE THUNDERSTORMS! THAT IS WHY HE IS ALSO CALLED THE THUNDER GOD. HE HAS LIT UP THE SKY WITH LIGHTNING BOLTS FORMED INTO THE NAMES OF HIS FRIENDS AND HOME. WITH THE PROGRAM ON THE NEXT PAGE YOU CAN CREATE THE SAME EFFECT WITH YOUR NAME AND THE NAMES OF YOUR FRIENDS.**



# Program 9

## Namestar

Input the following program in your computer and run it. Type in your first and last name, hit return and watch Thor turn your name into lightning bolts. To stop the program, hit any key.

Type in your friends' names or any names you like as well.

Load your 900 lines before typing this program.

```
100 REM NAMESTAR
110 GOSUB 960:GOSUB 900
120 MX=INT(SH/2)
130 MH=INT(SW/2)
140 INPUT "TYPE YOUR NAME: ";A$
150 IF LEN(A$)>=MX THEN A$=LEFT$(A$,MX-1)
160 GOSUB 900
170 FOR I=1 TO LEN(A$)
180 C$=MID$(A$,I,1)
190 V(1)=MX-I
200 V(2)=MX+I
210 H(1)=MH-I
220 H(2)=MH+I
230 FOR J=1 TO 2
240 VT=V(J):HT=MH:GOSUB 910
250 PRINT C$;
260 VT=MX:HT=H(J):GOSUB 910
270 PRINT C$;
280 FOR K=1 TO 2
290 VT=V(J):HT=H(K):GOSUB 910
300 PRINT C$;
310 NEXT K
320 NEXT J
330 NEXT I
340 GOSUB 940:IF KY$=NU$ THEN 160
350 END
```



*CONTINUE ON  
FOR MORE  
ADVENTURE!*





# MOLE MAN'S COMPUTER SABOTAGE!

MOLE MAN AND HIS SUBTERRANEANS ARE INVADING THE SURFACE WORLD! AND HE HAS SABOTAGED MR. FANTASTIC'S COMPUTER TO PREVENT IT FROM ACTIVATING THE WEAPONS SYSTEMS IN THE BAXTER BUILDING. THE COMPUTER ASKS FOR 10 NUMBERS WHICH IT COMBINES INTO THE ACCESS CODE. BUT WHEN REED INPUTS THEM, THE COMPUTER REJECTS THE SEQUENCE. IT JUST GOES INTO AN ENDLESS LOOP. **CAN YOU HELP?** RUN THE PROGRAM. TEST IT. THEN SEE IF YOU CAN FIND THE BUG THAT'S BUGGING REED!





## Program 10 Sabotage

Input the following program in your computer and run it. Enter ten numbers.

You will see that the program is an endless loop.

List the program and see if you can figure out why the program only deals with the first number you put in. The computer might give you a clue if you interrupt the program after it fills the screen.


When you've solved the problem, run the program again to get the access code Mr. Fantastic needs to activate the weapons system.

If you need help in debugging the program, consult page 91 in the Hint Section.

Load your 900 lines before typing this program.

```
100 REM SABOTAGE
110 GOSUB 900
120 PRINT "OPEN THE LOCK"
130 PRINT "ENTER THE 10 READINGS"
140 PRINT "ONE AT A TIME"
150 FOR I=1 TO 10
160 PRINT "ENTER READING ";I;
170 INPUT R(I)
180 NEXT I
190 PRINT
200 I=1
210 PRINT "THE SECRET COMBINATION IS:"
220 REM PRINT OUT 10 NUMBERS
230 PRINT INT(100*(SIN(R(I))));" ";
240 IF I=10 THEN 260
250 GOTO 230
260 PRINT:PRINT
270 PRINT "THATS ALL OF THEM"
280 END
```

46



47



**SPEAK,  
ANDROID,  
SPEAK...**



REED RICHARDS IS  
EXPERIMENTING WITH A  
NEW ANDROID. HE'S  
TRYING TO TEACH IT TO  
SPEAK. YOU CAN HELP  
RUN THE PROGRAM ON  
THE NEXT PAGE. BUT  
DON'T BE SURPRISED IF  
THE RESULTS ARE  
PRETTY RIDICULOUS!





# Program 11

## Android Speech

Input the following program in your computer and run it.

Type in a sentence in all capital letters, using ADJC for an adjective, NOUN for a noun, ADVB for an adverb, and VERB for a verb. Include some real words in your sentence.

Press return and see what silly sentences the computer comes up with.

Type STOP when you've finished.

Load your 900 lines before typing this program.

```
100 REM ANDROID SPEECH
110 GOSUB 900:GOSUB 960
120 M1=9:M2=9:M3=9:M4=9
130 RX=M1:GOSUB 930:N=RD
140 RX=M2:GOSUB 930:V=RD
150 RX=M3:GOSUB 930:AD=RD
160 RX=M4:GOSUB 930:AV=RD
170 DATA JOYSTICK,FLIPPER,TELEVISION,FISH,SHOE
180 DATA CARPET,COMPUTER,SKYSCRAPER,FROG
190 DATA HITS,RUNS,DROWNS,DRINKS,SHOUTS AT
200 DATA EATS,TURNS ON,FLATTENS,BLOWS UP
210 DATA SOGGY,FLIMSY,AWKWARD,INSANE
220 DATA HARD,INTENSE,INTELLIGENT,STUPID,PURPLE
230 DATA QUICKLY,CRAZILY,HESITANTLY,CARELESSLY,EMPTYLY
240 DATA WITHOUT FEAR,CEASELESSLY,OVER AND OVER,KINDLY
250 FOR I=1 TO M1:READ N$(I):NEXT I
260 FOR I=1 TO M2:READ V$(I):NEXT I
270 FOR I=1 TO M3:READ AD$(I):NEXT I
280 FOR I=1 TO M4:READ AV$(I):NEXT I
290 PRINT "TYPE SENTENCE"
300 INPUT A$
310 IF A$= "STOP" THEN END
320 A$=A$+" "
330 I=1
```



```
340 J=1
350 K$=MID$(A$,I+J,1)
360 IF K$=" " OR K$="." THEN W$=MID$(A$,I,J):I=I+J:J=1:GOTO 400
370 J=J+1
380 IF I+J>LEN(A$) THEN 480
390 GOTO 350
400 X$=RIGHT$(W$,4)
410 IF X$="NOUN" THEN W$=N$(N):N=N+1:IF N>M1 THEN N=1
420 IF X$="VERB" THEN W$=V$(V):V=V+1:IF V>M2 THEN V=1
430 IF X$="ADJC" THEN W$=AD$(AD):AD=AD+1:IF AD>M3 THEN AD=1
440 IF X$="ADVB" THEN W$=AV$(AV):AV=AV+1:IF AV>M4 THEN AV=1
450 LL=LL+LEN(W$)+1:IF LL>SW-1 THEN LL=LEN(W$)+1:PRINT
460 PRINT W$;" ";
470 GOTO 350
480 PRINT
490 GOTO 290
```



UNACCUSTOMED  
AS I AM TO PUBLIC  
SPEAKING...



# THE THING TALKS!

WHAT'S THIS? **THE THING IS MAKING A SPEECH?** WELL, IT SEEMS REED'S EXPERIMENT WITH THE ANDROID RUBBED OFF ON HIS MUSCULAR PAL. RUN THE PROGRAM ON THE NEXT PAGE. **THE RESULTS WILL BE HILARIOUS!**



# Program 12

## Executive

Input the following program in your computer and run it. Ask The Thing to write you an essay as many times as you like. Read the essay carefully and you too can impress your friend with double-talk.

Load your 900 lines before typing this program.

```
100 REM EXECUTIVE
110 GOSUB 960:GOSUB 900:PRINT "EXECUTIVE"
120 PRINT "*****"
130 DIM A$(12),N$(12),V$(12),P$(12)
140 DIM S1(5),S2(5),S3(5)
150 DATA STUBBORN,THOUGHTFUL,BASIC,SMART,FAST
160 DATA UNSWERVING,WISHY-WASHY
170 DATA HOT,RUMORED,WISE
180 DATA DOCUMENTATION,TALKING,THINKING,IMAGINATION
190 DATA ARGUMENT,COOPERATION,PROOF,COMMUNICATION,WAR
200 DATA SUPRISE
210 DATA CAUSES,ENCOURAGES,INCLUDES,SUBJUGATES,MESSES UP
220 DATA STEALS, INTERRUPTS, CONTINUES,DISTORTS,LIES ABOUT
230 DATA VAGUE IMPRESSIONS,TAMPERING,AMPLIFICATION,FALSEHOODS
240 DATA CLEAR-CUT COMPROMISE
250 DATA SUPPOSED ANSWERS, SILLY CONTRADICTION,SHOUTING
260 DATA EMBARRASSMENT, THE SORT-OF-SYNDROME
270 FOR I=1 TO 10:READ A$(I):NEXT I
280 FOR I=1 TO 10:READ N$(I):NEXT I
290 FOR I=1 TO 10:READ V$(I):NEXT I
300 FOR I=1 TO 10:READ P$(I):NEXT I
310 INPUT "SHALL I WRITE AN ESSAY ";Y$
320 GOSUB 900
330 IF Y$="Y" THEN 350
340 END
350 GOSUB 900:PRINT :PRINT:PRINT
360 GOSUB 480
370 PRINT A$(X(1));" ";N$(X(2));" ";V$(X(3));" ";P$(X(4));
380 PRINT ",TO SUCH A DEGREE THAT ";
390 GOSUB 480
400 PRINT A$(X(1));" ";N$(X(2));" ";V$(X(3));" ";P$(X(4));
```



```
410 PRINT ", A CASE WHICH MUST LEAD TO THE CONCLUSION THAT ";
420 GOSUB 480
430 PRINT A$(X(1));" ";N$(X(2));" ";V$(X(3));" ";P$(X(4))"." ;
440 GOSUB 480
450 PRINT "THEREFORE ";A$(X(1));" ";N$(X(2));", ALWAYS ";
460 PRINT " ";V$(X(3));" ";P$(X(4));"!"
470 PRINT:PRINT:GOTO 310
480 FOR Z=1 TO 4
490 RX=10:GOSUB 930:X(Z)=RD
500 NEXT Z
510 RETURN
```




# PUMPKIN BOMBS!

TRICK OR TREAT,  
WEB HEAD!



**OH NO! IT'S SPIDER-MAN'S  
OLD ENEMY, THE HOBGOBLIN!**

**WATCH OUT, SPIDEY**—THOSE  
JACK-O-LANTERNS ARE  
BOMBS! QUICK! HELP  
SPIDER-MAN BEFORE IT'S  
TOO LATE! YOU MUST CATCH  
THE PUMPKIN BOMBS  
BEFORE THEY HIT THE  
GROUND AND **EXPLODE!!**





# Program 13

## Pumpkin Bombs

Type the following program into your computer. Then run it. You must catch the bombs in the special bomb-proof basket that's running back and forth at the bottom of the screen. The bombs will drop from the letters at the top of the screen. If you hit "T" on the keyboard, the bomb will drop from the "T". Hit "B" and it will drop from the "B", etc. The object of the game is to figure out which key to hit—and when—so the bomb lands in the basket. A score of 20 or above will save Spidey's hide. It's all up to you. Good luck, computer ace.



Load your 900 lines before typing this program.

```
100 REM PUMPKIN BOMBS
110 GOSUB 900
120 DATA 1,1,2
130 READ X,S,V
140 GOSUB 960
150 W=SW-8:SZ=SH-V2
160 B$(1)=" ***** "
170 B$(2)="   ***   "
180 GOSUB 900
190 PRINT LEFT$(" A B C D E F G H I J K L M N O P Q R S ",W+4)
200 D=0
210 IF D=1 THEN 280
220 GOSUB 940:IF KY$=NU$ THEN 290
230 C=ASC(KY$):C=C+128*(C>127)
240 B=(C-ASC("A")+2)*2-2
250 IF B>W+5 THEN B=W+5
260 IF B<1 THEN B=1
270 D=1
280 V=V+1:IF V=SZ THEN 360
290 IF D=0 THEN 320
```



```


300 VT=V-1:HT=B:GOSUB 910:PRINT " ";
310 VT=V:HT=B:GOSUB 910:PRINT "O";
320 X=X+S:IF X>W THEN X=W:S=-1:K=K+1:IF K>15 THEN 450
330 IF X<1 THEN X=1:S=1
340 GOSUB 400
350 GOTO 210
360 IF ABS(X+3-B)<=2 THEN M=M+1:GOSUB 480:GOTO 390
370 HT=B:VT=V-1:GOSUB 910:PRINT " ";
380 VT=V:GOSUB 910:PRINT "@";
390 V=2:GOTO 200
400 FOR I=1 TO 2
410 VT=SZ-2+I:HT=X:GOSUB 910
420 PRINT BS(I);
430 NEXT I
440 RETURN
450 GOSUB 900
460 PRINT " YOU CAUGHT ";M;" BOMBS"
470 END
480 HT=SW/2:GOSUB 910:PRINT "CAUGHT!";:FOR WT=1 TO 500:NEXT:RETURN

```



# FIND THE BOMB!

**THE HOBGOBLIN HAS HIDDEN ANOTHER BOMB SOMEWHERE IN THIS WAREHOUSE. SPIDEY MUST FIND THE BOMB BEFORE IT GOES OFF. IT WON'T BE EASY—THE WAREHOUSE IS ALMOST A SQUARE MILE. BUT YOU CAN HELP! ENTER THE PROGRAM ON THE NEXT PAGE AND HELP SPIDER-MAN FIND THE BOMB—BEFORE IT FINDS YOU!**





# Program 14

## Find Bomb

Input the following program in your computer and run it. Using the letters and numbers on the screen as coordinates, try and find the bomb.

Enter the letter first, then the number.

After you enter the first coordinates, the computer will help you find the bomb.

Load your 900 lines before typing this program.

```
100 REM FIND BOMB
110 GOSUB 960
120 GOSUB 900
130 D$="....."
140 A=ASC("A")-1
150 VT=1:HT=4:GOSUB 910
160 FOR I=1 TO SK
170 PRINT CHR$(A+I);
180 NEXT I
190 FOR I=1 TO SK
200 VT=I+1:HT=1:GOSUB 910
210 PRINT I;
220 HT=4:GOSUB 910
230 PRINT LEFT$(D$,SK)
240 NEXT I
250 RX=SK:GOSUB 930
260 R1=RD
270 GOSUB 930:R2=RD
280 VT=SK+GP:HT=1:GOSUB 910
290 PRINT "WHERE'S THE BOMB?";
300 VT=SK+GP:HT=17:GOSUB 910
310 PRINT " ";
320 GOSUB 910
330 INPUT A$
340 X=ASC(LEFT$(A$,1))-A
350 Y=VAL(MID$(A$,2))
360 IF (X=R1)*(Y=R2)THEN 450
370 PRINT "THE BOMB IS LOCATED "
```



```

380 G=G+1:IF G=5 THEN 470
390 IF (Y>R2) THEN PRINT "NORTH";
400 IF (Y<R2) THEN PRINT "SOUTH";
410 IF (X>R1) THEN PRINT "WEST";
420 IF (X<R1) THEN PRINT "EAST";
430 PRINT " OF ";A$;" "
440 GOTO 280
450 GOSUB 900:PRINT "YOU FOUND THE BOMB"
460 END
470 GOSUB 900:PRINT "*****"
480 PRINT "*** BOOM!!! ***"
490 PRINT "*****"
500 END

```

---

# FIND BOMB CHANGES FOR SPECIFIC MACHINES

## 1. COMMODORE 64

```
970 SK=14:GP=4:RD=RND(-TI):RETURN
```

## 2. VIC-20

```
970 SK=10:GP=3:RD=RND(-TI):RETURN
```

## 3. APPLE

```
970 SK=14:GP=4:RETURN
```

## 4. IBM PC and PCjr

```
970 SK=14:GP=4:RANDOMIZE VAL(RIGHT$(TIME$,2)):RETURN
```

## 5. TRS 80 MODEL III and COLOR COMPUTER


```
970 SK=10:GP=3:RETURN
```



# GET THE NUMBER

**NOW WHAT INDEED! CAN A SUPER HERO GET NO REST?**

NOT TODAY. SPIDEY IS TRAPPED INSIDE THE WAREHOUSE. THE ONLY WAY HE CAN GET OUT IS TO COME UP WITH THE SECRET NUMBER THE COMPUTERIZED LOCK IS LOOKING FOR. AND HE HAS TO DO IT FIVE TIMES. YOU'D BETTER HELP—OR THIS COULD BE THE END FOR THE AMAZING SPIDER-MAN.





# Program 15

## Secret Sequence

Input the following program in your computer and run it. Enter any number from 1-10 to start a sequence.

Then study the number sequences to help Spider-Man find the secret number.


You must get five sequences right to defuse the bomb. If you fail, the bomb will explode.

Hint: You may need a calculator to help you.

Load your 900 lines before typing this program.

```
100 REM SECRET SEQUENCER
110 GOSUB 960:GOSUB 900
120 INPUT "ENTER A SEQUENCE START ";S1
130 SQ=S1+2
140 RX=3:GOSUB 930:R1=RD
150 RX=5:GOSUB 930:R2=RD
160 PRINT:PRINT
170 FOR I=1 TO 5
180 ON R1 GOSUB 300,310,320
190 PRINT SQ;" ";
200 NEXT I
210 ON R1 GOSUB 300,310,320
220 PRINT:PRINT
230 INPUT "NEXT NUMBER IS ";N
240 IF N=SQ THEN PRINT "YOU GOT IT!":C=C+1:GOTO 280
250 PRINT "WRONG!"
260 WR=WR+1:IF WR>=3 THEN PRINT "THE BOMB EXPLODED!":END
270 GOTO 130
280 IF C=5 THEN PRINT "BOMB DEFUSED":END
290 GOTO 120
300 SQ=SQ+R2:RETURN
310 SQ=SQ-R2:RETURN
320 SQ=SQ*R2:RETURN
```





## **SPIDEY'S SPECTACULAR EXIT**

**HOW WILL SPIDEY FINALLY  
MAKE HIS ESCAPE? INPUT  
THE PROGRAM ON THE NEXT  
PAGE AND WATCH WHAT  
HAPPENS.**



# Program 16

## Spidey's Good-Bye

Input the following program in your computer and run it.

Watch Spidey say "So long."

Load your 900 lines before typing this program.

```
100 REM SPIDEY'S GOODBYE
110 GOSUB 900:GOSUB 960
120 NM$="SO LONG BUNKY"
130 IF LEN(NM$)>40 THEN 160
140 NM$=NM$+"*" +NM$
150 GOTO 130
160 FOR I=1 TO SH-1
170 K=I
180 IF K>SH THEN K=SH:PRINT
190 FOR J=1 TO LEN(LEFT$(NM$,I))
200 VT=K:HT=SW-1-J:GOSUB 910
210 PRINT LEFT$(LEFT$(NM$,I),J);
220 NEXT J
230 NEXT I
240 FOR I=1 TO SH
250 RESTORE:VT=I:HT=1:GOSUB 910
260 WT=WU:GOSUB 920
270 READ A$:IF A$="END" THEN 310
280 VT=VT+1:IF VT>SH THEN 300
290 GOSUB 910:PRINT A$;
300 GOTO 270
310 NEXT
320 WT=WU:GOSUB 920
330 END
340 DATA " "
350 DATA " *** "
360 DATA " ***** "
370 DATA " ***** "
380 DATA " ***** "
390 DATA " ***** "
400 DATA " ' ' ' "
410 DATA " ' ' ' "
420 DATA " ' ' ' "
```



```
430 DATA " ' ' ' "
```

```
440 DATA " ' ' ' "
```

```
450 DATA " 'O' "
```

```
460 DATA " + "
```

```
470 DATA " X "
```

```
480 DATA "END"
```

---

SPIDEY'S GOOD-BYE  
CHANGES FOR SPECIFIC MACHINES

1. TRS 80 MODEL III

```
970 WU=50:RETURN
```

2. ALL OTHERS

```
970 WU=250:RETURN
```



1. A SMEX DWO ZOHŁOD LEIDO EV DWO HENBIDOL OXIABNOMD ZWABNOMD.

2. A BRUM DE ZDOUR DWO ZDIVV.

3. DWOŁO AZ MEDWAMK TULOTOGAR HUM TE DE ZDEB NO.

# THE TALKING CODE

WAIT A MINUTE, DAREDEVIL'S EXTRAORDINARY HEARING IS PICKING UP A MESSAGE FROM STILT-MAN. IS DAREDEVIL'S OLD ENEMY UP TO HIS USUAL TRICKS?

DAREDEVIL CAN HEAR HIM— BUT HE SEEMS TO BE TALKING IN CODE. IT'S UP TO YOU TO DECODE THE MESSAGE. THE PROGRAM ON THE NEXT PAGE WILL HELP.

AFTER YOU HAVE DECODED THE MESSAGE, HELP DAREDEVIL WRITE STILT-MAN A MESSAGE IN THE SAME CODE. **HEY, EVEN SUPER HEROS HAVE TO HAVE A LITTLE FUN, RIGHT?**





# Program 17

## Talking Code

Input the following program in your computer and run it. Then enter the coded message from the preceding page one line at a time. Be sure to follow the computer's instructions and use an asterisk to begin each line. Type STOP when you've finished.

Load your 900 lines before typing this program.

```
100 REM TALKING CODE
110 GOSUB 900
120 PRINT "TYPE STOP TO END PROGRAM"
130 PRINT "TO DECODE, START SENTENCE WITH A *"
140 PRINT "LIKE THIS: *DWO HUD."
150 PRINT:PRINT
160 DATA "UPHTOVKWAQSRNMEBXLZDIGYFJC"
170 DATA "ABCDEFGHIJKLMNOPQRSTUVWXYZ"
180 READ A$,B$
190 INPUT "TYPE A SENTENCE: ";S$
200 D=0
210 IF S$="STOP" THEN 380
220 C$=A$:P$=B$
230 IF LEFT$(S$,1)="*" THEN D=1:C$=B$:P$=A$
240 FOR I=1 TO LEN(S$)
250 L$=MID$(S$,I,1)
260 IF D THEN GOSUB 390:GOTO 290
270 IF L$<"A" OR L$>"Z" THEN 310
280 J = ASC(L$)-ASC("A")+1
290 IF J<1 OR J>26 THEN 310
300 L$=MID$(C$,J,1)
310 X$=X$+L$
320 NEXT I
330 PRINT
340 PRINT X$
350 PRINT
360 X$=""
370 GOTO 190
380 END
390 FOR J=1 TO 26
```



```
400 IF L$=MID$(P$,J,1) THEN 430
410 NEXT J
420 J=27
430 RETURN
```





# DOCTOR STRANGE

## AND THE COMPUTER GREMLIN

SOMETHING'S STRANGE WITH STRANGE. HE IS EXPERIMENTING WITH THE MAGIC OF COMPUTERS TOO, BUT HE CLAIMS THERE IS A GREMLIN INSIDE HIS MACHINE. EVERY TIME HE TRIES TO INPUT A BINARY NUMBER, A STRANGE CREATURE COMES ONTO THE SCREEN, GOBBLES THE NUMBER AND CHANGES IT BACK INTO A DECIMAL. RUN THE PROGRAM ON THE NEXT PAGE AND WATCH WHAT HAPPENS.

# Program 18

## Gremlin

Input the following program in your computer and run it. When the computer asks you for bits, enter a binary number—that's a base two number. It's all made up zeros and ones. Watch the creature gobble up the binary number and change it to a base-ten or decimal number.

Run the program as many times as you like, but press RUN for each play.

Load your 900 lines before typing this program.

```
100 REM GREMLIN
110 GOSUB 960
120 GOSUB 900
125 X=1
130 DATA " <"," =" ," "
140 FOR I=1 TO 3: READ A$(I):NEXT
150 INPUT "PLEASE FEED ME BITS ";B$
160 FOR I=1 TO LEN(B$)-1 : X=X*2:NEXT I
170 VT=INT(SH/2):HT=3:GOSUB 910:PRINT B$
180 FOR I=1 TO LEN(B$)
190 FOR J=1 TO 3
200 VT=INT(SH/2):HT=I:GOSUB 910:PRINT A$(J);
210 WT=WU:GOSUB 920
220 NEXT J
230 IF MID$(B$,I,1)<>"0" THEN T=T+X
240 X=INT(X/2)
250 VT=SH-4:HT=1:GOSUB 910:PRINT "I ATE ";T;" TASTY BITS!";
260 NEXT I
270 VT=SH-3:HT=1:GOSUB 910:PRINT "YUM! YUM! YUM!"
280 END
290 WT=WU:GOSUB 920
300 RETURN
```



GREMLIN  
CHANGES FOR SPECIFIC COMPUTERS

ALL COMPUTERS

970 WU=100:RETURN

# IRON MAN'S LOCK PROBLEM



IRON MAN MUST GET INTO STARK INTERNATIONAL'S SAFE, BUT SOMEONE HAS TAMPERED WITH THE LOCK. HE KNOWS THE COMBINATION. IT'S 12, 13, 18. BUT SOMETHING'S WRONG. **HAS THE COMBINATION BEEN CHANGED?** TURN THE PAGE TO ENTER THE PROGRAM INTO YOUR COMPUTER AND SEE IF YOU CAN HELP IRON MAN.



# Program 19

## Program Lock

Input the following program in your computer and run it. Enter the numbers from the preceding page. You will see that Iron Man has made a mistake.

You must figure out what the correct numbers are for the combination to open the safe.

The display will give you some clues or list the program and study it. The program listing will also give you a clue.

If you still need help, consult the Hint Section, Page 94.

Load your 900 lines before typing this program.

```
100 REM PROGRAM LOCK
110 GOSUB 900
120 FOR I=1 TO 15
130 PRINT "STARK INTERNATIONAL ";
140 X=I
150 NEXT I
160 PRINT
170 FOR I=1 TO 12
180 PRINT "TOP SECRET SAFE ";
190 Y=I
200 NEXT I
210 PRINT
220 FOR I=1 TO 19
230 PRINT "OPEN AT RISK ";
240 Z=I
250 NEXT I
260 PRINT:PRINT
270 PRINT "WHAT'S THE COMBINATION?"
280 PRINT "THREE NUMBERS #,#,#"
290 INPUT A,B,C
```

```
300 REM IS COMB CORRECT?
310 IF ((X=A)*(Y=B)*(Z=C)) THEN 350
320 FOR I = 1 TO 50
330 PRINT " WRONG!!!"
340 NEXT I:END
350 GOSUB 900:PRINT "THE LOCK HAS OPENED"
360 END
```





# DAREDEVIL'S TOUCH

# OF EVIL

DAREDEVIL'S SENSE OF TOUCH IS SO DEVELOPED THAT HE CAN DETECT THINGS THAT ORDINARY PEOPLE CAN'T SEE. HE RUBS HIS FINGERS OVER A RANSOM NOTE RECEIVED BY THE POLICE. **HE CAN TELL THAT SOME OF THE WORDS WERE WRITTEN AT A DIFFERENT TIME.** (THE INK FEELS DRIER.) HE CONCLUDES THAT THE WORDS ON THE PAGE ARE DUMMY WORDS—THERE ARE ONLY A FEW LETTERS THAT REALLY GIVE THE MESSAGE.

**TOO BAD, YOU DON'T HAVE DD'S POWERS.** YOU'LL HAVE TO DO THIS A SLOWER WAY. INPUT THE PROGRAM ON THE NEXT PAGE. START REMOVING LETTERS, KEEP ELIMINATING UNTIL YOU HAVE A 6-LETTER WORD THAT MAKES SENSE. GOOD LUCK.



# Program 20

## Letters Out

Input the following program and run it.

Remove one letter from the message at a time by hitting that key.

The computer will automatically remove the letters. Keep removing letters until you end up with only six letters left. If you have removed the right letters, the six remaining ones will spell the name of Daredevil's enemy. If you don't get the name the first time, run the program again.

Load your 900 lines before typing this program.

```
100 REM LETTERS OUT
110 GOSUB 900:GOSUB 960
120 READ S$:READ T$
130 S$=S$+T$
140 FOR I=1 TO LEN(S$)
150 K$=MID$(S$,I,1)
160 IF K$<>" " AND K$<>"." THEN CNT=CNT+1
170 NEXT
190 GOSUB 900
200 HT=1:VT=3:GOSUB 910
210 IF KT<>0 THEN PRINT KT;" ";KY$;"'S REMOVED":PRINT:GOTO 230
220 PRINT:PRINT
230 PRINT S$
240 IF CNT=0 THEN 460
250 HT=1:VT=10:GOSUB 910
260 PRINT "REMOVE?=";
270 GOSUB 940:IF KY$=NU$ THEN 270
280 HT=1:VT=3:GOSUB 910
290 PRINT "                                ";
300 HT=1:VT=10:GOSUB 910:KT=0
310 PRINT "REMOVING ";KY$;"                                "
320 FOR I=1 TO LEN(S$)
330 IF MID$(S$,I,1)<>KY$ THEN 430
340 CNT=CNT-1
350 S1$="":IF I=1 THEN 370
```



```
360 S1$=LEFT$(S$,I-1)
370 S2$="":IF LEN(S$)=I THEN 390
380 S2$=RIGHT$(S$,LEN(S$)-I)
390 S$=S1$+"-"+S2$
400 HT=1:VT=5:GOSUB 910:KT=KT+1
410 PRINT S$
420 WT=WV:GOSUB 920
430 NEXT
440 HT=10:VT=10:GOSUB 910:PRINT"  "
450 GOTO 200
460 HT=1:VT=SH-2:GOSUB 910
470 PRINT"FINISHED"
480 END
490 DATA "KIDNAPPING PAIN I CAN NIP IN BUD "
500 DATA "IF I FIND 100 GRAND IN THE BOWL IN CABIN."
```

---


LETTERS OUT  
CHANGES FOR SPECIFIC COMPUTERS

ALL MACHINES

970 WV=250:RETURN



# THE THING'S THINK TANK



ALL THIS PROGRAMMING HAS MADE THE THING WANT TO GET INTO THE COMPUTER AGE. AND HE'S BOUND AND DETERMINED TO MAKE SOMETHING POP UP ON THAT COMPUTER SCREEN. TURN THE PAGE TO FIND OUT WHAT KIND OF PROGRAM HE CREATES.



# Program 21

## Thing's a Poppin

Input the following program in your computer and run it.

Enter the number of kernels and watch what happens.

Load your 900 lines before typing this program.

```
100 REM THING'S A POPPIN'
110 GOSUB 960:GOSUB 900
120 B=SH-2
130 GOSUB 290
140 PRINT "HOW MANY KERNELS?":INPUT K
150 GOSUB 900
160 RX=SW:GOSUB 930:R1=RD
170 IF R1=1 THEN PP$="*"+MID$(A$,2):GOTO 200
180 IF R1=SW THEN PP$=MID$(A$,1,SW-1)+"*":GOTO 200
190 PP$=MID$(A$,1,R1-1)+"*"+MID$(A$,R1+1)
200 VT=B:HT=1:GOSUB 910
210 PRINT PP$;
220 RX=B-1:GOSUB 930:VT=RD:HT=R1:GOSUB 910
230 PRINT "*";
240 KC=KC+1
250 IF KC<K THEN 160
260 VT=B:HT=1:GOSUB 910
270 PRINT A$;
280 END
290 A$=""
300 FOR I=1 TO SW
310 A$=A$+"."
320 NEXT I
330 RETURN
```





# **SUPER HINT SECTION**



## General Hints



1. As you know, the programs in this book won't run correctly if they aren't typed into your computer exactly as they are in the listings. Nearly everyone makes one or two typing mistakes, so don't be surprised if your program doesn't run as it should the first time.

To check for mistakes, simply LIST the program. For a short program, just type LIST and hit RETURN (or ENTER or whatever your computer calls the key that enters commands). On longer programs, it's best to LIST a few lines at a time, so the program doesn't scroll up the screen and go past you. To do this, simply type something like LIST 100-170 and only those lines will appear on the screen. Then after you've checked those, you can check the next few lines—until you find your mistake.



2. When you are finished with one program and about to start another, it's a good idea to type NEW. That will clear out any leftovers from the previous program that could mess up your new program.

3. Remember to use all capital letters to type in the programs or to input data. Some computers have a CAP LOCK key (or LOCK). If you press it, all the letters will appear in caps.

4. Some of these programs will run on forever unless you interrupt or break them.

All computers have a way to interrupt a program. On the Commodore, you press the RUN/STOP key. On the IBM it's two keys: the CTL and BREAK. You hold them down at the same time. On the APPLE, you hold down the CONTROL key and press the letter C. The Radio Shacks have a BREAK key. If your computer keyboard is different, just consult your owner's manual.

Note: You may be able to think of different ways to achieve the same effects as these programs. Often these programs were not written in the fastest or easiest way. That's because the programs were designed to run on several computers. Besides, no two programmers ever write programs exactly the same way. If you want to customize or change these programs, go right ahead. That's part of the fun of computers! Good luck.

**TURN THE PAGE  
FOR EVEN MORE  
HELPFUL HINTS!**





# Hints for Specific Programs

## Program 1 **Bombs Away**

Bombs Away is an action game. The bombs appear on your screen in one of four places. You must shoot them before they disappear. To shoot a bomb, type "U" for Up, "D" for Down, "L" for Left, or "R" for Right. The value of "S" in line 970 determines how quickly or slowly the bombs come and go. You can make it a smaller number to make the game more challenging. Your score is presented at the end of the game.

## Program 2 **Guess Who**

Guess Who is a guessing game. It's a computer version of hangman, without the hanging! You get to make only five mistakes before guessing the secret word(s) or you lose! To make the game more fun, we gave you the words in the program listing in code. You can add your own words and phrases to play the game with your friends. To code them, make each letter one higher than it is in the real word. That means "CAT" would be coded as "DBU". Put the coded words in a DATA statement numbered 520 or higher. Also change the value of RX in line 120 to be equal to the number of phrases in your list. Each phrase is separated by a comma. Here's an example: 120 RX = 3: GOSUB 930:GOSUB 960

```
530 DATA CMBDL DBU,TMZ GPY
```

These statements will add the phrases BLACK CAT and SLY FOX to the available list. For more fun, have friends code up phrases for you so you can play.

## Program 3 **Crasher**

Crasher is an arcade-style game. The object is to crash through the hole in the fence. If you crash into the fence



you start at the bottom again. Use the letters "J" and "K" to move left or right. The crusher moves up automatically. The hole in the fence moves randomly on the fence. Programming challenge: Try to modify the program so that it asks if you want to play again after you clear the fence successfully.

## Programs 4 and 5

### **Kingpin's Decoder and Encoder**

The message from Kingpin that Spider-Man intercepted was put in code by this Encoder program. Here's how it works:

First, it makes a mirror image of your message. "THE CAT EATS" becomes "STA E TAC EHT." Next a certain number that you choose, from 1 to 26, is added to each letter of the backwards code. If you choose a secret number of "1" then an "A" becomes a "B" and so on. In this example about the cat, the final result is "TUBF UBD FIU" for a secret number of "1". Try to encode the same message several times with different numbers and see how the results are similar and how they are different.

In order to read the message, you must decode it using Decoder. Decoder is a lot like Encoder. Only a few lines are different. Look at the listing carefully. Can you find which lines they are?

## Program 6

### **Marquee**

Marquee is a fun program that puts your name or any message you like in print that moves just like a theater marquee. You can use this program to make a computerized greeting card for someone special. You can make the message go faster or slower by changing the value of "WU" in line 970. The bigger the number, the slower the marquee will go.



## Program 7

### **Type Training**

Type Training is a special program that challenges you to type as quickly as possible. It times you by counting the number of times the keyboard is checked (or "polled") before the correct letter is typed. This means that the timer units are different for every microcomputer. Keep track of your scores. If you practice, your typing and your scores should get better and better.

## Program 8

### **Robot**

Robot is an animation program with a bug. You can save the mission and the robot by changing the program to reverse its direction after it gets the secret plans. How do you do it? Look at line 150. The variable "D" is the return direction. At line 290 it is added to the horizontal position (HT) of the robot's return trip. Of course if "D" is a positive number, the robot will always keep moving to the right, right into the trap. How can you make the robot go back to the left? What number could "D" be so that HT got smaller instead of bigger? Be sure that your robot takes only one step at a time. That's right, change "D" to "minus one."

## Program 9

### **Namestar**

Namestar is a text graphics program. It makes an expanding star by printing the letters of a name in different directions on the screen. Here's a programming challenge: Change the program so that it reads five names from a DATA statement instead of asking for a name at the beginning of the program.



## Program 10

### Sabotage

Sabotage tests your ability to debug a program. Someone has tampered with the program. They have removed a crucial piece of code. What is it? Where should it be? The same number is printed over and over. You entered the readings into the 10 elements of the array named "R." Look at line 230. Is it processing every element of that array? Where does the variable "I" get incremented? I starts at the value one. Can it ever reach 10? Try to add line 235 so that I is incremented before the IF-THEN statement.

Try changing 240 to  $I = I + 1$  and see what happens.

## Program 11

### Android Speech

Android Speech is a sentence construction program. To use it you must provide the sentence structure. The computer puts in the actual words. You supply the articles (the, a) and conjunctions (and, or, but) and the program supplies the rest of the words. Wherever you want the computer to insert a noun, type "NOUN." Do the same for adjectives ("ADJC"), adverbs ("ADVB"), and verbs ("VERB"). Here is an example of the kind of sentence you might enter:

THE ADJC NOUN VERB THE ADJC NOUN VERB ADVB  
AND THE NOUN VERB THE ADJC NOUN ADVB.

The silly response will be made up of the words in the DATA statements.

You can substitute your own favorite nouns and adjectives into the DATA statements. Keep the same number of words, or change the values of M1, M2, M3, and M4 to the number of words you do include. (M1 is noun count, M2 is verb count, M3 is adjective count, M4 is adverb count.) If you use more than 10 words in a category you may need to dimension other variables, so it's easier to keep the count at ten or less.



## Program 12

### **Executive**

Executive is an essay writer. It combines randomly chosen phrases with an essay skeleton to form essays. We do not recommend that you use these essays as homework for your English class. On the other hand, the essays are pretty impressive sounding. You can change the phrases. Leave the same number of entries in the DATA statements but substitute new, similar words. Substitute nouns for nouns, verbs for verbs, and so on. Enjoy the essays!

## Program 13

### **Pumpkin Bombs**

Pumpkin Bombs is a fun game where you try to catch a falling bomb in a basket by typing the letter the bomb should drop from. One thing you can easily change about this program is the basket. As long as you leave it the same size, you can make the basket out of any letters or symbols you like. How about using your initials or your name? Be sure to put the new characters only where there are asterisks (\*) in the DATA statements in lines 160 and 170. The spaces should stay where they are.

## Program 14

### **Find Bomb**

Find Bomb is a matrix puzzle. You try to find the bomb by guessing its coordinates. You have only a few chances, so think it out. The best strategy for winning a game like this is to try to divide the search in half each time.

## Program 15

### **Secret Sequence**

Secret Sequence asks you to determine the next number in a sequence. For this program we just used add,



subtract, and multiply. You can see these sequence makers in lines 300, 310, and 320. Can you see a way to change the sequences? Try changing the multiply sequence to be two times the previous number plus one. The formula would be:

$$SQ = SQ * R2 + 1$$

You can try other formulas, too. Use your imagination and see what happens!

## Program 16

### **Spidey's Good-Bye**

Spidey's Good-Bye is a great example of text animation.

Mostly it's just a lot of fun to watch. Try changing the phrase "SO LONG BUNKY!" to something else. Watch what happens. Try changing the pattern (but not the size) of the parachute. Put your initials in the middle.

## Program 17

### **Talking Code**

Talking Code is a code that works on the principle of alphabetic substitution. Each character of the alphabet is replaced by another: an A becomes a U, B becomes P, and so on. In this case we chose our alphabet so that the code can usually be pronounced like a strange language. Also this code program is designed so that it can both encode and decode messages. If you want to decode a message, all you need to do is precede the coded message with an asterisk (\*). Remember that you can end the program by typing 'STOP' when asked.

## Program 18

### **Gremlin**

Gremlin eats binary numbers. A binary number is a number made up of ones and zeros. Each position in the binary number is a power of two. That means that the



number furthest to the right is the one's place, the one to the left of it is the two's place, then the four's place, the eight's place, and so on. A one in a position means add the amount, a zero means don't add it. This means that 10010 is  $16 + \text{no } 8 + \text{no } 4 + 2 + \text{no } 1$ . Total 18. Isn't it nice the Gremlin will do all that complicated stuff for you? Try some other binary numbers. Feed the Gremlin, and he will work for you!

## Program 19

### **Program Lock**

Program Lock requires you to study the code to find the hidden combination. The combination is embedded in the program itself. If you have trouble finding it, look very carefully at the variables "X", "Y", and "Z". Where do they change? The combination is 12, 15, 19. Can you see now how it was hidden? Can you change the program so that the combination also changes?

## Program 20

### **Letters Out**


Letters Out takes letters that you choose out of a message. You can use it to hide special messages in DATA statements and then remove the extra letters to recover the messages. If you pull out the right letters in our example, you'll discover the villain behind the kidnapping — the Owl.

## Program 21

### **Thing's a Poppin'**

Thing's a Poppin' is just a very simple graphics program. You can change it by changing the character used for "popcorn" in all the lines where the asterisk is used.





---

## **SUBJECT INDEX TO TYPES OF PROGRAMS**

### **CODES:**

Kingpin's Decoder	Page 25
Kingpin's Encoder	Page 27
Talking Code	Page 70

### **DEBUGGING PUZZLES:**

Program Lock	Page 78
Robot	Page 38
Sabotage	Page 46

### **GAMES:**

Bombs Away	Page 14
Crasher	Page 22
Find Bomb	Page 60
Pumpkin Bombs	Page 57
Robot	Page 38
Sabotage	Page 46
Secret Sequence	Page 63
Type Training	Page 34

### **NUMBER GAMES:**

Gremlin	Page 74
Secret Sequence	Page 63

### **SPECIAL EFFECTS:**

Gremlin	Page 74
Marquee	Page 30
Namestar	Page 42
Spidey's Good-Bye	Page 66
Thing's a Poppin	Page 84

### **WORD GAMES:**

Android Speech	Page 50
Executive	Page 54
Guess Who	Page 19
Letters Out	Page 81



## About the Authors

**Richard Guaraldo** is an electrical engineering graduate from Drexel University in Philadelphia, PA, and an avid microcomputer enthusiast. His love of micros began when he purchased his first APPLE II in 1980. He quickly became involved with the Maryland Crab Apples, a local APPLE users group, where he served as newsletter editor, secretary, and president. His interests have expanded to CP/M, the COMMODORE 64, and the IBM-PC. In addition to this book, he has written several educational packages for the COMMODORE 64 that are used in the Anne Arundel County, Maryland, Public School System. He resides in Crofton, Maryland, with his wife and daughters.

**Susan M. Zakar** is a computer systems analyst/programmer living in Gambrills, Maryland. She has a bachelor's degree in foreign languages from Western Washington State College. In 1978, she purchased her first personal computer (an Apple II) and became so enchanted with the world of microcomputers that she eventually made it her career. She has programmed on many different kinds of microcomputers, in FORTRAN, BASIC, Pascal, and assembly language. She has contributed to, and acted as editor to several microcomputer user group newsletters. She also chairs a microcomputer user group at work, and she has created programs for the successful computer/adventure paperback series *Micro Adventures™*. Her interest in programming for young people was inspired by her two-year-old, for whom she hopes to write programs which 'fascinate and entertain.'



# Your computer will turn you into a Super Hero!

That's right! Your computer makes you part of the Marvel Super Heroes'™ world of adventure. Program your computer to help the Marvel Super Heroes defeat the forces of evil. **21 super programs make the challenge fun!**

**IRON MAN MUST KEEP ENEMY MISSILES FROM HITTING YOUR SCHOOL! PROGRAM YOUR COMPUTER WITH REPULSOR RAYS AND HELP IRON MAN SAVE THE DAY!**

**THE GRAY GARGOYLE HAS GOT CAPTAIN AMERICA TRAPPED! ONLY YOUR COMPUTER CAN HELP HIM ESCAPE!**



**SPIDER-MAN LEAVES A SECRET MESSAGE FOR HIS ENEMY THE KINGPIN. USE YOUR COMPUTER TO BREAK THE CODE AND FIND OUT WHAT SPIDEY HAS TO SAY.**

**PROFESSOR X HAS CREATED A TEST TO MEASURE RESPONSE SPEED. TRY IT ON YOUR COMPUTER AND SEE HOW YOU COMPARE WITH THE X-MEN.**

So turn on your computer and get ready to load it with codes, computer puzzles, games, and graphics!

## **PROGRAMS CAN BE RUN ON ALL THESE PERSONAL COMPUTERS!**

**APPLE II • APPLE IIe • APPLE II+ • APPLE IIc • COMMODORE 64  
VIC-20 • IBM PC • IBM PC JR. • RADIO SHACK COLOR COMPUTER  
RADIO SHACK TRS-80™ MOD 3**

ISBN: 0-917657-06-3



DONALD I. FINE, INC.